[6.4] Gamma, log-gamma and factorial functions

The gamma function is one of many 'special functions' that occur as the result of integrations, limits, sums and products. The gamma function is defined as

$$\Gamma(z) = \int_0^\infty t^{(z-1)} e^{-t} dt \qquad \text{Re}(z)>0$$

The gamma function is defined for all real and complex numbers except for negative integers. The gamma function can also be used to define factorials of negative numbers and non-integers, since

$$\Gamma(n+1) = n!$$

The CAS occasionally returns results including the gamma function, but there is no built-in function to calculate it. Here's a function to calculate gamma:

```
gamma(z)
Func
© Γ(Z) by Stirling's formula  ©M.Dave1
If real(floor(z))=z:Return when(z>Ø,(z-1)!,undef)
Local n,x,y
when(real(z)<Ø,1-z,z)→x
1Ø-min(floor(abs(x)),1Ø)→n
If  n≠Ø:x+n→x
approx(ℯ^(⁻x)*x^(x-Ø.5)*√(2*π)*polyEval({Ø.ØØØ83949872Ø872Ø9,⁻5.17179Ø9Ø826Ø6ᴇ⁻ØØ5,⁻Ø.ØØØ
59216643735369,6.9728137583659ᴇ⁻ØØ5,Ø.ØØØ784Ø3922172ØØ7,⁻Ø.ØØØ229472Ø936214,⁻Ø.ØØ26813271
6Ø4938,Ø.ØØ3472222222222,Ø.Ø83333333333333,1},1/x)*when(n=Ø,1,∏(1/(x-k),k,1,n)))→y
approx(when(real(z)<Ø,π/(sin(z*when(sin(π)=Ø,π,18Ø)))/y,y))
EndFunc
```

This function works for all real and complex arguments. The accuracy is near full machine precision, except for *very* large negative arguments.

Since the gamma function is similar to the factorial function, the result overflows the calculator floating point range for relatively small arguments. For example, *gamma(z)* returns infinity for z>450. This limitation can be overcome by using a function that returns the natural log of the gamma function, instead of the gamma function itself. The log-gamma function is:

```
lngamma(z)
Func
© lnΓ(Z) by asymptotic series ©M.Dave1
If fPart(z)=Ø and z<1:Return undef
Local n,x,y
when(real(z)<Ø,1-z,z)→x
1Ø-min(floor(abs(x)),1Ø)→n
If n≠Ø:x+n→x
approx((x-Ø.5)*ln(x)-x+Ø.5*ln(2*π)+polyEval({⁻Ø.ØØØ595238Ø952381,Ø.ØØØ79365Ø79365Ø79,⁻Ø.Ø
Ø27777777777778,Ø.Ø83333333333333},x^(⁻2))/x+when(n=Ø,Ø,ln(∏(1/(x-k),k,1,n))))→y
approx(when(real(z)<Ø,ln(π/(sin(z*when(sin(π)=Ø,π,18Ø))))-y,y))
EndFunc
```

The program author, Martin Daveluy, has these additional comments:

*These two series use asymptotic series combined with the recurrence formula (gamma(Z+1) = Z*gamma(Z)) for Z<10 to keep full precision and the reflection formula (gamma(Z)*gamma(1-Z) = pi/(sin(pi*Z))) to extend domain to the entire complex plane. Note that the Gamma Stirling's fomula is obtained by this LnGamma formula. The Stirling's coefficients are obtained by collecting X power of the Maclaurin series for e^X (1+X+(X^2)/2!+... with X= LnGamma_asymptotic_series) to reach higher precision.*

With a gamma function, it is possible to write a factorial function for non-integer and complex arguments:

```
factrl(xx)
func
©(x)factorial, complex & non-integer arguments
©1jan00/dburkett@infinet.com
when(imag(xx)=0 and fpart(xx)=0 and xx≤450,xx!,e^(math\lngamma(xx+1)))
Endfunc
```

If the input argument is a real integer less than 450, the result is found with the built-in factorial function is used, otherwise the log-gamma function is used. Note that *lngamma()* is installed in the *math* folder.

(credit to Martin Daveluy)